

Aufgabe 3 (Auswertungsstrategie):

(26 Punkte)

Gegeben sei das folgende Haskell-Programm:

```
doubleElements :: [Int] -> [Int]
doubleElements [] = []
doubleElements (x:xs) = x : x : doubleElements xs

firstN :: Int -> [Int] -> [Int]
firstN n [] = []
firstN n (x:xs) =
  if n > 0 then x : firstN (n-1) xs else []

listSum :: [Int] -> Int
listSum [] = 0
listSum (x:xs) = x + listSum xs
```

Die Funktion `doubleElements` berechnet eine Liste, welche aus einer gegebenen Liste hervorgeht, indem jedes Element dieser Liste noch einmal hinter sich selbst eingefügt wird. Zum Beispiel wertet der Ausdruck `doubleElements [1,2,3]` zu `[1,1,2,2,3,3]` aus. Die Funktion `firstN` gibt zu jeder Liste `xs` den längsten Anfang `ys` von `xs` zurück, sodass `ys` höchstens eine gegebene Anzahl an Elementen besitzt. Der Ausdruck `firstN 2 [1,2,3]` wertet beispielsweise zu `[1,2]` aus, während hingegen `firstN 5 [1,2,3]` zu `[1,2,3]` auswertet. Die Funktion `listSum` bildet die Summe über alle Elemente einer Liste vom Typ `[Int]`. Wird `listSum [1,2,3]` aufgerufen, so ergibt sich 6.

Geben Sie alle Zwischenschritte bei der Auswertung des Ausdrucks

```
listSum (firstN (1+0) (doubleElements [1+0, 1+0]))
```

an. Unterstreichen Sie vor jedem Auswertungsschritt den Teil des Ausdrucks, der als Nächstes an seiner äußersten Position ausgewertet wird. Um Platz zu sparen können Sie hierbei `dE`, `fN` und `lS` statt `doubleElements`, `firstN` und `listSum` schreiben.

Hinweise:

- Beachten Sie, dass Haskell eine Leftmost-Outermost Lazy Auswertungsstrategie besitzt. Allerdings sind Operatoren wie `*`, `>` und `-`, die auf eingebauten Zahlen arbeiten, strikt, d. h. hier müssen vor Anwendung des Operators seine Argumente vollständig ausgewertet worden sein (wobei zunächst das linke Argument ausgewertet wird).

Lösung:

```

listSum (firstN (1+0) (doubleElements [1+0,1+0]))
→ listSum (firstN (1+0) ((1+0):(1+0):doubleElements [1+0]))
→ listSum (if 1+0 > 0 then (1+0):firstN ((1+0) - 1) ((1+0):doubleElements [1+0]) else [])
→ listSum (if 1 > 0 then (1+0):firstN (1 - 1) ((1+0):doubleElements [1+0]) else [])
→ listSum (if True then (1+0):firstN (1 - 1) ((1+0):doubleElements [1+0]) else [])
→ listSum ((1+0):firstN (1 - 1) ((1+0):doubleElements [1+0]))
→ (1+0) + listSum (firstN (1-1) ((1+0):doubleElements [1+0]))
→ 1 + listSum (firstN (1-1) (1:doubleElements [1+0]))
→ 1 + listSum (if 1-1 > 0 then 1:firstN (1-1-1) (doubleElements [1+0]) else [])
→ 1 + listSum (if 0 > 0 then 1:firstN (0-1) (doubleElements [1+0]) else [])
→ 1 + listSum (if False then 1:firstN (0-1) (doubleElements [1+0]) else [])
→ 1 + listSum []
→ 1 + 0
→ 1

```

Aufgabe 6 (Listen in Haskell):

(4+5+5+5+5 = 24 Punkte)

Seien x , y und z ganze Zahlen vom Typ `Int` und seien xs und ys Listen der Längen n und m vom Typ `[Int]`. Welche der folgenden Gleichungen zwischen Listen sind richtig und welche nicht? Begründen Sie Ihre Antwort. Falls es sich um syntaktisch korrekte Ausdrücke handelt, geben Sie für jede linke und rechte Seite auch an, wie viele Elemente in der jeweiligen Liste enthalten sind und welchen Typ sie hat.

Beispiel: Die Liste `[x,y],ys` hat den Typ `[[Int]]` und enthält 2 Elemente.

- `((x:[ys]) : []) : [] = (([x] ++ ys) : [] : []) ++ [[]]`
- `(x:[y]):[xs] = [[x] ++ [y]] ++ [xs]`
- `((x:[y]):[z]++ys):[] = x:y:z:ys`
- `(x:ys):[] ++ [xs] = (x:ys++xs):[[]]`
- `(x:y:([z]++ys)):xs = [x,y,z]:ys ++ xs`

Hinweise:

- Hierbei steht `++` für den Verkettungsoperator für Listen. Das Resultat von `xs ++ ys` ist die Liste, die entsteht, wenn die Elemente aus `ys` — in der Reihenfolge wie sie in `ys` stehen — an das Ende von `xs` angefügt werden.

Beispiel: `[1,2] ++ [1,2,3] = [1,2,1,2,3]`

- Falls linke und rechte Seite gleich sind, genügt **eine** Angabe des Typs und der Elementzahl.

Lösung:

- Der Teilausdruck `x:[ys]` des linken Ausdrucks ist nicht typkorrekt, da `[ys]` den Typ `[[Int]]` besitzt, jedoch wird hier ein Wert vom Typ `[Int]` erwartet, da `x::Int` gilt.
Der rechte Ausdruck ist eine Liste vom Typ `[[Int]]` und enthält 3 Elemente.
- Beide Ausdrücke repräsentieren dieselbe Liste der Länge 2 vom Typ `[[Int]]`, nämlich `[x,y],xs`

- c) Der linke Ausdruck entspricht der Liste `[[x,y],[z]++ys]` der Länge 1 vom Typ `[[[Int]]]`.
 Der rechte Ausdruck repräsentiert eine Liste mit $3+m$ Elementen vom Typ `[Int]`, nämlich `[x,y,z]++ys`.
 Beide Ausdrücke entsprechen also verschiedenen Listen.
- d) Beide Ausdrücke sind vom Typ `[[Int]]` und besitzen die Länge 2. Jedoch enthält die rechte Liste im Gegensatz zur Linken den Wert `[] :: [Int]` als zweites Element, während hingegen das zweite Element der linken Liste `xs :: [Int]` ist. Weiterhin ist das erste Element der linken Liste `x:ys :: [Int]` und das erste der rechten Liste `x:ys ++ xs :: [Int]`. Die Listen sind also verschieden.
- e) Beide Ausdrücke sind korrekt getypt und entsprechen Listen vom Typ `[[Int]]`. Die linke Liste hat Länge 2, während die rechte Liste die Länge 3 besitzt.
 Beide Ausdrücke entsprechen daher verschiedenen Listen.

Aufgabe 9 (Programmieren in Haskell):

(7+6+10+9+11+7 = 50 Punkte)

Implementieren Sie alle der im Folgenden beschriebenen Funktionen in **Haskell**. Geben Sie jeweils auch die Typdeklarationen an. Sie dürfen die Listenkonstruktoren `[]` und `:` (und deren Kurzschreibweise), `True` und `False`, Werte des Typs `Int`, die Listenkonkatenation `++`, Vergleichsoperatoren wie `<=`, `==`, `...`, boolesche Funktionen wie `&&`, `||`, `not` und die arithmetischen Operatoren `+`, `*`, `-` verwenden, aber **keine** weiteren vordefinierten Funktionen. Schreiben Sie ggf. Hilfsfunktionen, um sich die Lösung der Aufgaben zu vereinfachen. Sie dürfen jederzeit Hilfsfunktionen aus vorherigen Teilaufgaben verwenden, auch wenn Sie diese nicht selbst implementiert haben.

- a) **symmetricDifference xs ys**
 Diese Funktion berechnet die symmetrische Differenz zweier Listen `xs :: [Int]` und `ys :: [Int]`. Die symmetrische Differenz ist eine Liste `zs :: [Int]`, welche alle Elemente enthält, die entweder nur in `xs` oder nur in `ys` enthalten sind. Die Reihenfolge der Elemente innerhalb der resultierenden Liste ist hierbei unerheblich. Wenn ein Wert sowohl in `xs` als auch in `ys` vorkommt, darf er nicht in **symmetricDifference xs ys** auftreten, auch wenn er unterschiedlich oft in `xs` und `ys` vorkommt.
 Die Auswertung von **symmetricDifference** `[1,1,2,4,5]` `[2,2,3,4,6]` kann beispielsweise die Liste `[1,1,5,3,6] :: [Int]` liefern.
- b) **powerlist xs**
 Berechnet eine Liste `ps :: [[Int]]` aller Teillisten von `xs :: [Int]`. Für jede Liste in `ps` soll die Reihenfolge der Listenelemente dieselbe sein wie die Reihenfolge der entsprechenden Elemente in `xs`. Die Reihenfolge der Elemente von `ps` selbst ist hierbei unerheblich. Mehrfach vorkommende Werte in `xs` können auch in den Teillisten entsprechend oft auftreten. Wenn l die Länge von `xs` bezeichnet, so ist 2^l die Länge von **powerlist xs**.
 Die Auswertung von **powerlist** `[1,2,3]` kann beispielsweise die Liste `[[], [3], [2], [2,3], [1], [1,3], [1,2], [1,2,3]]` der Länge 8 ergeben. Der Ausdruck **powerlist** `[2,2]` kann hingegen beispielsweise zur Liste `[[], [2], [2], [2,2]]` ausgewertet werden.
- c) **permutations xs**
 Diese Funktion soll zu einer Liste `xs :: [Int]` eine Liste aller Permutationen von `xs` berechnen. Eine Permutation von `xs` ist hierbei eine Liste `ps :: [Int]`, welche dieselben Elemente wie `xs` enthält, jedoch kann die Reihenfolge der Elemente in `ps` von der Reihenfolge der Elemente in `xs` abweichen. Die Reihenfolge der Elemente von **permutations xs** kann hierbei beliebig sein. Wenn `xs` eine Liste der Länge l ist, so ist **permutations xs** von der Länge $l!$. Falls `xs` die leere Liste ist, so darf sich die Funktion beliebig verhalten.
 Der Ausdruck **permutations** `[1,2,3]` könnte damit zu der Liste `[[1,2,3], [2,1,3], [3,2,1], [1,3,2], [3,1,2], [2,3,1]]` vom Typ `[Int]` auswerten.

Die folgenden Teilaufgaben beziehen sich auf eine Datenstruktur, welche Graphen repräsentiert. Mathematisch ist ein Graph ein Tupel $\mathcal{G} = (\mathcal{V}, \mathcal{E})$, wobei \mathcal{V} eine Menge von Knoten (engl. *vertices*) und $\mathcal{E} \subseteq \mathcal{V}^2$ eine Menge von Kanten (engl. *edges*) ist, welche die Knoten untereinander verbinden. Die betrachteten Graphen sind gerichtet, d.h., eine Kante (a, b) bedeutet, dass Knoten b von Knoten a erreicht werden kann, jedoch nicht unbedingt auch umgekehrt. Wir nehmen im Folgenden an, dass jeder Knoten mindestens eine eingehende oder ausgehende Kante besitzt.




Abbildung 1: Durch die Liste `testGraph = [(1,2),(2,3),(3,1),(4,5),(3,4)] :: [(Int,Int)]` repräsentierter Graph.

Abb. 1 ist eine graphische Repräsentation des Graphen $(\{1, 2, 3, 4, 5\}, \{(1, 2), (2, 3), (3, 1), (4, 5), (3, 4)\})$. Im Folgenden werden Graphen in Haskell als Liste vom Typ `[(Int,Int)]` ihrer Kanten dargestellt. Der in Abb. 1 abgebildete Graph kann als Liste `testGraph = [(1,2),(2,3),(3,1),(4,5),(3,4)] :: [(Int,Int)]` seiner Kanten in Haskell dargestellt werden.

d) `nodes es`

Gegeben eine Liste `es :: [(Int,Int)]` von Kanten, berechnet diese Funktion eine Liste vom Typ `[Int]` aller im Graph enthaltenen Knoten. Die berechnete Liste soll *keine Duplikate* enthalten. Die Reihenfolge ist irrelevant.

Beispielsweise könnte `nodes testGraph` zu der Liste `[1,2,3,4,5] :: [Int]` auswerten.

e) `existsPath es a b`

Diese Funktion berechnet für eine gegebene Liste `es :: [(Int,Int)]` von Kanten und zwei Knoten `a, b :: Int` einen Wahrheitswert vom Typ `Bool`, der angibt, ob der Knoten `b` vom Knoten `a` aus mithilfe der Kanten aus der Liste `es` erreicht werden kann.

Die Ausdrücke `existsPath testGraph 1 3`, `existsPath testGraph 1 5` und `existsPath testGraph 5 5` berechnen hierbei beispielsweise den Wahrheitswert `True`, während hingegen die Aufrufe `existsPath testGraph 5 1`, `existsPath testGraph 5 4` und `existsPath testGraph 4 3` zu `False` auswerten. Für jeden Knoten `a` gilt hierbei, dass `existsPath es a a` zu `True` auswertet, da man den Knoten `a` immer von sich selbst aus über einen Pfad aus 0 Kanten erreichen kann.

Hinweise:

- Überlegen Sie, wie die Liste der Kanten des Graphen in einem rekursiven Aufruf geeignet modifiziert werden kann, um Terminierung bei zyklischen Graphen sicherzustellen.

f) `isConnected es`

Diese Funktion berechnet, ob ein durch eine Liste von Kanten `es :: [(Int,Int)]` dargestellter Graph zusammenhängend ist. Ein Graph heißt zusammenhängend, wenn für alle Knoten `a` und `b` der Knoten `b` von `a` aus erreichbar ist, d.h., `existsPath es a b` ist `True` für alle Knoten `a` und `b`.

Zum Beispiel wertet `isConnected testGraph` zu `False` und `isConnected ((5,1):testGraph)` zu `True` aus.

Lösung: _____

a) `remove :: Int -> [Int] -> [Int]`

`remove _ [] = []`

`remove e (x:xs) = if e==x then remove e xs else x:remove e xs`

`symmetricDifference :: [Int] -> [Int] -> [Int]`

```

symmetricDifference xs ys = difference xs ys ++ difference ys xs
  where
    difference :: [Int] -> [Int] -> [Int]
    difference xs [] = xs
    difference xs (y:ys) = difference (remove y xs) ys

b) powerlist :: [Int] -> [[Int]]
powerlist [] = [[]]
powerlist (x:xs) = powerlist xs ++ insertFront (powerlist xs)
  where
    insertFront :: [[Int]] -> [[Int]]
    insertFront [] = []
    insertFront (p:ps) = (x:p):insertFront ps

c) permutations :: [Int] -> [[Int]]
permutations [] = [[]]
permutations [x] = [[x]]
permutations (x:xs) = helper (permutations xs)
  where
    insertEverywhere :: [Int] -> [Int] -> [[Int]]
    insertEverywhere ps [] = [ps++[x]]
    insertEverywhere ps (e:es) =
      (ps++(x:e:es)):insertEverywhere (ps++[e]) es
    helper :: [[Int]] -> [[Int]]
    helper [] = []
    helper (l:ls) = insertEverywhere [] l ++ helper ls

d) testGraph = [(1,2),(2,3),(3,1),(4,5),(3,4)] :: [(Int,Int)]
nodes :: [(Int,Int)] -> [Int]
nodes xs = removeDuplicates (allNodes xs)
  where
    allNodes :: [(Int,Int)] -> [Int]
    allNodes [] = []
    allNodes ((a,b):es) = a:b:allNodes es
    removeDuplicates :: [Int] -> [Int]
    removeDuplicates [] = []
    removeDuplicates (x:xs) = x:removeDuplicates (remove x xs) -- a)

e) existsPath :: [(Int,Int)] -> Int -> Int -> Bool
existsPath g a b = helper [] a b g
  where
    helper :: [(Int,Int)] -> Int -> Int -> [(Int,Int)] -> Bool
    helper _ a b [] = a == b
    helper us a b ((x,y):es)
      | a == x = existsPath (us++es) y b || helper us a b es
      | otherwise = helper ((x,y):us) a b es

f) isConnected :: [(Int,Int)] -> Bool
isConnected g = checkPairs vs vs
  where
    vs :: [Int]
    vs = nodes g
    checkPairs :: [Int] -> [Int] -> Bool
    checkPairs [] _ = True
    checkPairs (_:xs) [] = checkPairs xs vs
    checkPairs (x:xs) (y:ys) = existsPath g x y && checkPairs (x:xs) ys
  
```